DOCUMENT NO. R6500 N12
JUNE 1979

Rockwell

R6500 Microcomputer System APPLICATION NOTE

A CRT Monitor or TV Interface for AIM 65

PURPOSE

The twenty-column printer and display on the AIM 65 is sufficient for most AIM 65 applications, but there are occasions when a longer line length, more lines per display or a separate video display is desired. The hardware and software described in this note offer a method of directing the display to a standard TV set, a modified TV set or a CRT monitor requiring a composite video input.

REQUIRED EQUIPMENT

The following equipment is required to build and trouble shoot the hardware, and to assemble the software into PROM.

1. Dual trace oscilloscope – DC to 10 MH
2. High impedance input multimeter
3. PROM programmer
4. Assembler
5. Schematic for TV
6. See Parts List

HARDWARE DESCRIPTION

Figure 1 is the block diagram for the interface circuit. All necessary timing and control signals are provided by the 6845 CRTC chip, IC1. The CRT controller chip provides the refresh addresses (MA0-MA13) to retrieve display characters from the refresh RAM and the row address so the character generator (IC2) can define the dot pattern for each row of each character to be displayed. The CRTC also provides the video timing (Hsyn, Vsyn) and display enable. These timings are programmable by writing into various registers of the chip. As designed, the registers are accessed by first writing the register number into memory location $8000 and then the desired value for that register into $8001 (see Figure 2).

Under control of the CRTC, the character generator (IC2) converts the ASCII code and row address into a seven-bit pattern which is transferred to an internal shift register. This bit pattern is then clocked out serially to produce the dot pattern shown in Figure 3. Timing and load for the shift register are provided by the gated oscillator (IC8 and related circuitry). This gated oscillator is clocked by the $\overline{\phi 2}$ signal.



Figure 1. DISPLAY CONTROLLER INTERFACE BLOCK DIAGRAM

REG. #	REG FILE	CALCULATION	PROGRAMMED VALUE DECIMAL	HEX
R0	H. TOTAL	64 X 1=64 USEC	64-1=63	$3F
R1	H. DISPLAYED	40 X 1=40 USEC	40	$28
R2	H. SYNC POSITION	50 X 1=50 USEC	50	$32
R3	H. SYNC WIDTH	02 X 1=02 USEC	02	$02
R4	V. TOTAL	23 X 704=16.18 MS	23-1=22	$16
R5	V. TOTAL ADJUST	08 X 64=512 USEC	08	$08
R6	V. DISPLAYED	16 X 704=11.26 MS	16	$10
R7	V. SYNC POSITION	19 X 704=13.30 MS	19	$13
R8	INTERLACE MODE			$00
R9	MAX SCAN LINE ADDR		11-1=10	$0A
R10	CURSOR START RASTER			$4A
R11	CURSOR END RASTER		11	$0B
R12	START ADDRESS (H)			$00
R13	START ADDRESS (L)			$00
R14	CURSOR (H)			$00
R15	CURSOR (L)			$00
R16	LIGHT PEN (H)			
R17	LIGHT PEN (L)			

CHARACTER ROW PERIOD = 11 X 64 = 704 USEC
VERTICAL PERIOD = (23 X 704) + (8 X 64) = 16.69 MS
TYPICAL 40 X 16 SCREEN FORMAT INITIALIZATION OF CRTC

Figure 2. CRTC REGISTER SUMMARY TABLE



Figure 3. DOT MATRIX CHARACTERS ON A RASTER SCAN

The AIM 65 R6502 processor communicates with the CRTC over the data bus via the AIM 65 expansion connector by reading or writing into the 18-register file of the CRTC. The AIM 65 accesses the refresh memory using IC3, IC4, and IC5 to multiplex the address bus and IC9 and IC10 to enable the data buffers of the display RAM. Since the AIM 65 has fixed clock cycles, a transparent refresh memory configuration is used. The CRTC accesses the memory during Ø2 while the processor accesses the memory during Ø2. The refresh memory acts as any other memory in the AIM 65 memory map. This method imposes no hardware burden on the AIM 65, and the refresh memory can be used for any other purpose when the display is not in use.

The remainder of the circuit generates the composite video input to a monitor, RF modulator or modified TV set. Note the optional reverse video circuitry enclosed in dotted lines on the schematic diagram. If this option is not selected, connect IC2-2 to IC11-12, connect IC11-11 to ground and connect IC11-13 to IC13-2.

SOFTWARE DESCRIPTION

Figure 4 is the flowchart for the software. The program may be assembled anywhere in memory; however, it is preferable to use a starting location above $AFFF if the CRT monitor or TV display is to be used as the principal display. With this program located above $AFFF the AIM 65 can single-step through some other user program in lower memory without also single-stepping through the video interface program. The listing given at the end of this application note starts at location $D800. This area of memory is assigned to the optional assembler. If the assembler is in place, a starting address below $CE34 should be assigned. Consult the memory map in the AIM 65 User's Guide (p. 7-45) when selecting a starting location. Note that locations $A000-$AFFF should not be used, except as noted below.

The program first initializes all the registers in the CRTC and changes two bytes of AIM 65 RAM ($A406 and $A407) to direct the display output to the CRT controller subroutine rather than to the display-driver. This initialization routine need only be called at power-on, if power to the AIM 65 is interrupted, or the contents of locations $A406 and $A407 are changed.



Figure 4. SOFTWARE FLOWCHART

The program decodes the control commands listed in Figure 5. If the character is none of these commands, it is written into the display refresh RAM. If the optional reverse video circuitry has been incorporated in the hardware, the command "switch video" (ASCII $07) will cause all following characters to be displayed in reverse video, until another "switch video" command is given.

COMMAND DESCRIPTION	ASCII CODE	KEY
CARRIAGE RTN & LINE FEED	0D, 8D	RETURN
CLEAR SCREEN	18	CONTROL X
ERASE TO END OF SCREEN	03	CONTROL C
HOME CURSOR	01	CONTROL A
SCROLL UP	11	CONTROL Q
CURSOR RIGHT	0F	CONTROL O
CURSOR LEFT	09	CONTROL I
CURSOR UP	0B	CONTROL K
CURSOR DOWN	0C	CONTROL L
SWITCH VIDEO	07	CONTROL G

Figure 5. COMMAND SUMMARY

The program is designed to allow various line lengths with a minimum of software change. The listing sets the maximum line length at 40 characters. This may be changed by changing LINMAX to the desired line length and changing SCRMAX to 16 times the desired line length. The second byte of TABL must be changed to the desired line length. The software handles scrolling and wrap-around.

NOTE

Line lengths over 32 characters require modification of a standard TV set. See "TV Modifications".

Should the user want to design his own software, it will be necessary to recover the delete character control signal (ASCII $7F). The software in the AIM 65 monitor will decrement the character pointer for the AIM 65 display and printer, but it will not echo the delete control character to the CRTC. The STORE routine (program lines 0195 and following) illustrates one method of recovering the control character.

CHECK-OUT

Do not attempt to connect the CRT controller to a TV set or CRT monitor until the following tests have been performed.

Inspect all power supply and ground connections within the CRT controller board and between the board and the AIM 65. Measure the DC current drawn by the AIM 65 without the CRT controller board connected, then measure the current drawn by the AIM 65 with the controller board connected. The difference should be no more than 0.6 Amps.

Using the "M" and "/" commands at several locations between $9000 and $93FF check that the AIM 65 is reading and writing into the display RAM.

For purpose of check out it is desirable to assemble the CRT controller software in the low RAM. Set the program counter on the AIM 65 to the beginning of the initialization routine. Using the "G" command run the controller program. If unexpected results occur in the following procedures, the problem may be in the software or the hardware of the CRT controller.

Check Vsync, IC1, pin 40, for a pulse every 16.66 milliseconds and Hsync, IC1, pin 39, for a pulse every 64 microseconds. Next check the output from the gated oscillator, IC2, pin 21, and the load pulses, IC2, pin 23. These two signals are illustrated in Figure 6. The clock should have narrow positive pulses between the load pulses.



Figure 6. CHARACTER GENERATOR INPUT SIGNAL WAVEFORMS

Check the output of the internal shift register in the character generator, IC2, pin 2, for pulses indicating a dot pattern output. Adjust variable registers P1 and P2 to obtain the wave pattern illustrated in Figure 7 at the video out jack.

When the above tests and adjustments are successfully completed connect the video output from the CRT controller board to the video input jack on the CRT monitor, modified TV set (see discussion below) or RF modulator connected to the TV antenna terminals.

Re-initialize the interface software. Enter the AIM 65 Editor using the "E" command and assign sufficient buffer space for the maximum number of characters per display. Fill the edit buffer and therefore the monitor screen with characters.



Figure 7. HSYNC, VSYNC AND VIDEO COMBINED

It may be necessary to center the picture. This will require some experimentation. The picture may be moved upward by decrementing the Vsync position register on the CRTC. It may be moved to the right by decrementing the Hsync position register. With the monitor software in RAM decrement or increment the third and eighth byte in TABL of the software. Re-initialize the software after each change and note the position of the display. When the display is centered the revised values for these bytes should be included in the final code.

A FEW REFINEMENTS

Two 4K blocks of addresses are decoded on the AIM 65 board: $\overline{CS8}$ and $\overline{CS9}$. If the interface is built according to the diagram, $\overline{CS9}$ may be used to access only 1K ($9000-$93FF) of the 4K block. Likewise, $\overline{CS8}$ may be used to access only two locations ($8000 and $8001) of the second 4K block. Figure 8 illustrates a way of decoding the addresses to make more efficient use of memory space. Pin 9 of the 74LS155 should be connected to pin 3 of IC11 on the controller board and pin 12 of IC12 should be connected to pin 25 of IC1 on the controller board. If this



Figure 8. ADDRESSES DECODER

Note the Use of the Two Spare Gates in IC12.

option is selected, the software must be changed to reflect the reassignment of the CRTC registers. The first three equates should be changed to

RS	= $AC00
CURLOW	= $AC01
CURHIG	= $AC01

NOTE

CURLOW and CURHIG are the same address because of the way in which the registers in the 6845 are addressed.

OPERATING NOTES

Once the initialization routine has been run the display output will be directed to the video monitor or TV set. The prompt (^) characters and other meaningless outputs which will appear on the AIM 65 display are normal and should be disregarded. The special characters and the lower case characters shown in Figure 9 may be displayed by writing into the appropriate location in the display RAM ($9000 and above).

If it is desirable to change back to the AIM 65 display from the CRT display, change memory location $A402 to 00 and press the RESET button.

TV MODIFICATIONS

WARNING

TV sets retain high voltages even after being disconnected. Take proper precautions!

Important Note: Do not attempt to modify a TV set with a "hot-chassis" (one side of the power line tied to ground) or an AC-DC set. SUCH SETS GREATLY INCREASE THE DANGER OF SEVERE SHOCK. If it is necessary to use such a TV set, it must be isolated from the power source with a transformer or the video signal from the CRT controller board must be isolated from the TV set with an optical isolator. Most transistor and IC TV sets have power transformers which eliminate this problem. BE SURE!



Figure 9. ASCII CHARACTER SET (7-BIT CODE)

Bit 8 is reverse video.

In order to achieve line lengths up to 50 characters per line the composite video signal from the CRT controller board must be fed directly to the input of the video amplifier on the TV set. This is due to the narrow band width of the RF and IF amplifiers in both black and white and color TV sets. Also smaller TV sets tend to frame the picture more tightly than larger ones. This may prevent line lengths of over 40 characters even in TV sets which have been modified.

It is impossible to give the exact circuit which will allow direct video input for every TV. The user must consult the schematic for the TV set being modified. The user may also be helped in the modification by consulting the TV Typewriter Cookbook by Don Lancaster or the Television Engineering Handbook by Donald Fink.

Figure 10 illustrates the generalized circuit for the necessary modification to the TV set. When the TV circuit is broken at the input to the video amplifier a DC bias will, in most cases, have to be provided to this amplifier. P1 and P2, when properly adjusted, should in many cases, provide the necessary voltage levels. In other cases a 1-5 μf tantalum capacitor, with the positive side connected to the video output of the CRT controller board, along with a properly designed resistive bias circuit may be required.

The usual video source impedance is either 70 or 100 ohms. Coaxial cable of one of these impedances should be used for the composite video signal from the CRT controller board to the video amplifier of the TV set. All lengths of coax should be kept as short as possible. Note that the shield should be grounded as near the video amplifier of the TV as possible and at the video output of the CRT controller board. The shield should not be grounded at any other point. If a phone jack is used as illustrated in Figure 10, it should not be grounded. Ground loops will prevent the proper operation of the interface.



Figure 10. DIRECT VIDEO MODS USING A PHONE JACK

PARTS LIST

Item	Part	Qty	Description	
1	IC1	1	MC6845	CRTC (controller)
2	IC2	1	CG5004L-1	Character Generator (STD Microsystem) and Shift Register
3	IC3, IC4, IC5	3	SN74LS157	Address Multiplexer
4	IC6, IC7	2	2114	Display RAM (1K x 8)
5	IC8, IC12	2	7404	Hex Inverters
6	IC9, IC10	2	8T28	Data Transceivers
7	IC11	1	7402	Quadruple 2-Input NOR Gates
8	IC13	1	7433	NOR Gates with open collectors
9	IC14, IC16	2	7474	Dual D Flip-Flops
10	IC15	1	7486	Exclusive-OR Gates
11	D1, D2, D3	3	1N4148	Or equivalent computer diode
12	R1, R2, R3, R4, R5, R6	6	3.3K	1/4 Watt Carbon Film Resistor
13	R7, R8, R9	3	1K	1/4 Watt Carbon Film Resistor
14	R10	1	680Ω	1/4 Watt Carbon Film Resistor
15	R11	1	220Ω	1/4 Watt Carbon Film Resistor
16	R12	1	100Ω	1/4 Watt Carbon Film Resistor
17	P1	1	2K	Potentiometer (Adjust sync level)
18	P2	1	100Ω	Potentiometer (Adjust signal amplitude)
19	Q1	1	2N2222	Transistor
20	C1	1	220 pf	Polystyrene
21	C2	1	47 pf	Polystyrene
22	C3	1	270 pf	Polystyrene

```
LINE #  LOC      CODE        LINE

0002  0000                 ;EQUATES FOR THIS PROGRAM
0003  0000                 LINMAX =40              ;MAXIMUN # OF CHAR IN A LINE
0004  0000                 SCRMAX =0640            ;CHAR IN THE SCREEN (40 X 16)

0006  0000                 ;REGISTERS AND POINTERS
0007  0000                 RS      =$8000          ;REG SELECT (ADDR ALL OTHER REGS)
0008  0000                 CURLOW =$8001           ;CURSOR REG LOW (6845)
0009  0000                 CURHIG =$8001           ;CUR REG HIGH (6845)
0010  0000                 REFRAM =$9000           ;REFRESH RAM (9000-93FF)
0011  0000                 CURPO2 =$A415           ;DISP PNTR FOR CHR IN A LINE (AIM)
0012  0000                 CURSOR =$F0             ;CURSOR FOR INDIREC (9000-93FF)
0013  0000                 CURP2  =$F2             ;CURPO2 PRIME (UP TO 60)
0014  0000                 CURPO  =$F3             ;PNTR IN A LINE (0-40)
0015  0000                 REVER  =$F4             ;REVERSE VIDEO FLAG
0016  0000                 TEMP   =$A42D
0017  0000                 DILINK =$A406           ;DISPL OR TV LINKAGE
0018  0000                 DIBUFF =$A438           ;DISPLAY BUFFER (AIM)
0019  0000                 PHXY   =$EB9E           ;PUSH X,Y INTO STACK
0020  0000                 PLXY   =$EBAC           ;PULL X,Y FROM STACK

0022  0000                       *=$D800
0023  D800                 ;INITIALIZE DEFAULT VALUES (# OF CHR IN A LINE,
0024  D800                 ;# OF LINES , ETC.)
0025  D800  A2 00              LDX #0
0026  D802  86 F4              STX REVER          ;CLEAR REVERSE VIDEO
0027  D804  8E 00 80     INIT   STX RS             ;SET UP REG SELECT TO 1 OF 16 REGS
0028  D807  BD 2F D8            LDA TABL,X         ;SET REGISTERS TO VALUE FROM TABLE
0029  D80A  8D 01 80            STA RS+1           ;A0=1
0030  D80D  E8                  INX
0031  D80E  E0 10              CPX #16
0032  D810  D0 F2              BNE INIT
0033  D812                 ;CHANGE THE VECTOR ADDRESS FROM DISPLAY TO
0034  D812                 ;TV SET , WE NEED TO DO IT WITH A PROGRAM
0035  D812  A9 5C              LDA #<OUTTV        ;TRANSFER VECTOR TO TV
0036  D814  8D 06 A4           STA DILINK
0037  D817  A9 D8              LDA #>OUTTV
0038  D819  8D 07 A4           STA DILINK+1
0039  D81C  20 14 D9           JSR HOMECU         ;HOME CURSOR
0040  D81F  A9 18              LDA #$18           ;CLEAR SCREEN COMMAND
0041  D821  20 5C D8           JSR OUTTV
0042  D824                 ;CLEAR NEXT UNUSED LINE OF RAM SO SCROLL DOESN'T
0043  D824                 ;FLASH THIS LINE
0044  D824  A2 28              LDX #LINMAX
0045  D826  A9 20              LDA #$20           ;PUT SPACES
0046  D828  9D 80 92     CLR    STA SCRMAX+REFRAM,X ;NEXT LINE AFTER SCREEN
0047  D82B  CA                  DEX
0048  D82C  10 FA              BPL CLR
0049  D82E  00                  BRK                ;RETURN TO MONITOR
0050  D82F  3F            TABL  .BYT 63,40,50,02,22,08,16,19
0050  D830  28
0050  D831  32
0050  D832  02
0050  D833  16
0050  D834  08
0050  D835  10
```

```
LINE #  LOC   CODE        LINE

0050   D836  13
0051   D837  00                    .BYT 00,10,$4A,11,00,00,00,00
0051   D838  0A
0051   D839  4A
0051   D83A  0B
0051   D83B  00
0051   D83C  00
0051   D83D  00
0051   D83E  00
0052   D83F                ;OUTPUT <CR> ,<LF> TO T.V.
0053   D83F  A9 27         CRLFTV LDA #LINMAX-1   ;MOVE CURSOR ALL THE WAY RIGHT
0054   D841  38                   SEC
0055   D842  E5 F3                SBC CURPO
0056   D844  18           CRTV0   CLC
0057   D845  65 F0                ADC CURSOR       ;CURSOR + (LINMAX - CURPO)
0058   D847  85 F0                STA CURSOR
0059   D849  90 02                BCC CRTV1
0060   D84B  E6 F1                INC CURSOR+1
0061   D84D  A9 00         CRTV1  LDA #0
0062   D84F  8D 15 A4             STA CURPO2       ;CLEAR DISPLAY POINTER (AIM)
0063   D852  85 F2                STA CURP2
0064   D854  85 F3                STA CURPO        ;CLR DISP PNTR
0065   D856  20 64 D9     CURRIG  JSR STCH1        ;INCR CURSOR ONE MORE
0066   D859  4C B9 D8             JMP ENTCH1       ;SCROLL IF NEEDED

0068   D85C                ;----------ENTRY POINT----------
0069   D85C                ;OUTPUT A CHARACTER TO T.V.
0070   D85C  48           OUTTV   PHA              ;SAVE A
0071   D85D  20 9E EB             JSR PHXY         ;SAVE X,Y
0072   D860  A4 F1                LDY CURSOR+1     ;TEST FOR RANGE
0073   D862  C0 94                CPY #>REFRAM+4   ;IS CURSOR BELOW MAXIMUM ? (1K RAM)
0074   D864  B0 57                BCS HOME         ;NO,  HOME CURSOR & RTN
0075   D866  C0 90                CPY #>REFRAM     ;IS CURSOR ABOVE MINIMUM ?
0076   D868  90 53                BCC HOME         ;NO, HOME CURSOR
0077   D86A  A8                   TAY              ;SAVE CHARACTER
0078   D86B  C9 8D                CMP #$8D         ;<CR> WITH MSB=1 ?
0079   D86D  F0 D0                BEQ CRLFTV       ;YES ,SO <CR>,<LF>
0080   D86F  C9 20                CMP #$20         ;IS IT A COMM OR CHR TO BE ENTERED
0081   D871  B0 43                BCS ENTCHR
0082   D873                ;IT IS A COMMAND , NOW WHICH ONE
0083   D873  C9 0D                CMP #$0D         ;<CR> ?
0084   D875  F0 C8                BEQ CRLFTV
0085   D877  C9 18                CMP #$18         ;CLEAR SCREEN ?
0086   D879  F0 22                BEQ CLRSCR
0087   D87B  C9 03                CMP #$03         ;ERASE TO END OF SCREEN ?
0088   D87D  F0 26                BEQ EOS
0089   D87F  C9 01                CMP #$01         ;HOME CURSOR ?
0090   D881  F0 3A                BEQ HOME
0091   D883  C9 11                CMP #$11         ;SCROLL UP ?
0092   D885  F0 41                BEQ SCROLL
0093   D887  C9 0F                CMP #$0F         ;MOVE CURSOR RIGHT ?
0094   D889  F0 CB                BEQ CURRIG
0095   D88B  C9 0C                CMP #$0C         ;MOVE CURSOR DOWN ?
0096   D88D  F0 58                BEQ CURDOW
```

```
LINE #  LOC     CODE        LINE

0097   D88F   C9 0B              CMP #$0B        ;MOVE CURSOR UP ?
0098   D891   F0 63              BEQ CURUP
0099   D893   C9 09              CMP #$09        ;MOVE CURSOR LEFT ?
0100   D895   F0 6E              BEQ CURLEF
0101   D897   C9 07              CMP #$07        ;SET OR RESET REVERSE VIDEO ?
0102   D899   F0 70              BEQ REVIDE
0103   D89B   D0 19              BNE ENTCHR      ;ANY OTHER KEY (EVEN MSB=1)



0105   D89D                ;**** CLEAR SCREEN ****
0106   D89D   20 14 D9     CLRSCR JSR HOMECU      ;HOME CURSOR
0107   D8A0   20 1F D9            JSR SPACES      ;ENTER SPACES TO END OF SCREEN
0108   D8A3   F0 18               BEQ HOME        ;PUT CURSOR AT TOP & EXIT

0110   D8A5                ;**** ERRASE TO THE END OF SCREEN ****
0111   D8A5   A5 F1        EOS    LDA CURSOR+1    ;SAVE CURSOR
0112   D8A7   48                  PHA
0113   D8A8   A5 F0               LDA CURSOR
0114   D8AA   48                  PHA
0115   D8AB   20 1F D9            JSR SPACES      ;CLEAR TILL END OF SCREEN
0116   D8AE   68                  PLA
0117   D8AF   85 F0               STA CURSOR      ;RESTORE CURSOR
0118   D8B1   68                  PLA
0119   D8B2   85 F1               STA CURSOR+1
0120   D8B4   D0 0A               BNE HOM1        ;EXIT

0122   D8B6                ;**** ENTER A CHR & SCROLL IF NEEDED ****
0123   D8B6   20 27 D9     ENTCHR JSR STORE       ;STORE CHR IN RAM & INC CURSOR
0124   D8B9   F0 0D        ENTCH1 BEQ SCROLL      ;SCREEN OVERFLOW SO SCROLL THEM
0125   D8BB   D0 03        ENTCH2 BNE HOM1        ;SCREEN DIDNT OVERFLOW SO RTN

0127   D8BD                ;**** HOME CURSOR ****
0128   D8BD   20 14 D9     HOME   JSR HOMECU      ;HOME THRU SUBR
0129   D8C0   20 77 D9     HOM1   JSR TRANSF      ;CURSOR TO CURSOR REG (6845)
0130   D8C3   20 AC EB            JSR PLXY        ;RESTORE REGS
0131   D8C6   68                  PLA
0132   D8C7   60                  RTS

0134   D8C8                ;**** SCROLL ONE LINE UP ****
0135   D8C8   20 14 D9     SCROLL JSR HOMECU      ;HOME CURSOR
0136   D8CB   A0 28        SCR1   LDY #LINMAX     ;ADD OFFSET TO INDEX
0137   D8CD   B1 F0               LDA (CURSOR),Y  ;TRANSFER CHARACTERS
0138   D8CF   20 60 D9            JSR STCH0
0139   D8D2   D0 F7               BNE SCR1
0140   D8D4   18                  CLC
0141   D8D5   A9 02        SCR2   LDA #>SCRMAX    ;CURSOR TO BEGINNING OF LAST LINE
0142   D8D7   09 90               ORA #>REFRAM    ;START ADDRESSING FROM 9000 ON RAM
0143   D8D9   85 F1               STA CURSOR+1
```

```
LINE #  LOC     CODE        LINE

0144  D8DB  A9 58           LDA #<SCRMAX-LINMAX
0145  D8DD  85 F0           STA CURSOR
0146  D8DF  B0 DF           BCS HOM1          ;FINISH
0147  D8E1  20 1F D9        JSR SPACES        ;CLR LAST LINE
0148  D8E4  38              SEC
0149  D8E5  B0 EE           BCS SCR2

0151  D8E7                 ;**** MOVE CURSOR DOWN ****
0152  D8E7  A5 F0        CURDOW LDA CURSOR       ;ADD LINMAX TO CURSOR
0153  D8E9  18              CLC
0154  D8EA  69 28           ADC #LINMAX       ;MAXIMUM # OF CHAR IN A LINE
0155  D8EC  85 F0           STA CURSOR
0156  D8EE  90 03           BCC CURD2
0157  D8F0  20 68 D9        JSR STCH2         ;INC CURSOR HIGH
0158  D8F3  4C B9 D8     CURD2  JMP ENTCH1       ;SCROLL IF NEEDED

0160  D8F6                 ;**** MOVE CURSOR UP ****
0161  D8F6  A5 F0        CURUP  LDA CURSOR       ;SUB LINMAX  TO CURSOR
0162  D8F8  38              SEC
0163  D8F9  E9 28           SBC #LINMAX       ;MOVE UP ONE LINE
0164  D8FB  85 F0           STA CURSOR
0165  D8FD  B0 C1           BCS HOM1          ;UNDER FLOW OF PAGE
0166  D8FF  20 C5 D9        JSR DECR1         ;DECRE UPPER CURSOR
0167  D902  4C BB D8        JMP ENTCH2        ;SCROLL OR RTN

0169  D905                 ;**** MOVE CURSOR TO THE LEFT ****
0170  D905  20 BD D9     CURLEF JSR DECREM       ;DECREM CURSOR ,CURSOR+1
0171  D908  4C BB D8        JMP ENTCH2        ;SCROLL OR RTN

0173  D90B                 ;**** SET OR RESET REVERSE VIDEO ****
0174  D90B  A5 F4        REVIDE LDA REVER        ;EXCLUSIVE OR VIDEO FLAG
0175  D90D  49 FF           EOR #$FF          ;WE ONLY CARE FOR MSB
0176  D90F  85 F4           STA REVER
0177  D911  4C C0 D8        JMP HOM1
```

```
LINE # LOC      CODE        LINE

0179  D914                ;SUBROUTINES
0180  D914  A9 00         HOMECU LDA #0           ;SET CURSOR TO BEGINNING
0181  D916  85 F0                STA CURSOR
0182  D918  85 F3                STA CURPO        ;CLR DISP PNTR
0183  D91A  A9 90                LDA #>REFRAM
0184  D91C  85 F1                STA CURSOR+1
0185  D91E  60                   RTS

0187  D91F  A9 20         SPACES LDA #$20         ;PUT BLANK
0188  D921  20 60 D9             JSR STCH0
0189  D924  D0 F9                BNE SPACES
0190  D926  60                   RTS

0192  D927                ;FIRST CHECK IF DELETE WAS KEYED
0193  D927                ;IF THERE WAS A "JSR PSLS" (DELETE SUBROUTINE)
0194  D927                ;LOCATIONS FROM STACK POINTER PLUS 6,7 SHOULD BE $E7F2
0195  D927  BA            STORE  TSX              ;GET STACK POINTER
0196  D928  BD 07 01             LDA $107,X      ;SEE 7 PLACES UP FROM STACK POINTER
0197  D92B  C9 E7                CMP #$E7
0198  D92D  D0 07                BNE STCHR       ;IT IS NOT A DELETE CHARACTER
0199  D92F  BD 06 01             LDA $106,X      ;CHECK LOWER ADDRESS
0200  D932  C9 F2                CMP #$F2
0201  D934  F0 58                BEQ DELETE      ;YES ADDRESS SO DELETE ONE CHAR
0202  D936                ;STORE A CHAR IN RAM (9XXX) & CHECK FOR CURSOR
0203  D936  98            STCHR  TYA             ;CHARACTER WAS SAVED ON Y
0204  D937                ;WRAP CURPO2 AROUND 20 TO BE ABLE TO RECEIVE DELETES
0205  D937  AC 15 A4             LDY CURPO2      ;DONT LET CURPO2 >=20 CHR
0206  D93A  C8                   INY
0207  D93B  C0 14                CPY #20         ;CURPO2 >=20 ?
0208  D93D  90 02                BCC *+4         ;YES ,INCREM CURPO2
0209  D93F  A0 13                LDY #19         ;NO ,RESET TO 19
0210  D941  8C 15 A4             STY CURPO2
0211  D944                ;MAINTAIN DISPLAY BUFFER FOR EDITOR
0212  D944  A4 F2                LDY CURP2       ;IF > 60 DONT PUT IT ON DISBUFFER
0213  D946  C0 3C                CPY #60         ;MAX # OF CHR IN DISPL BUFFER (AIM)
0214  D948  B0 05                BCS STCH00
0215  D94A  99 38 A4             STA DIBUFF,Y    ;EDITOR & M-COMMAND USE THIS BUFFER
0216  D94D  E6 F2                INC CURP2
0217  D94F                ;WRAP AROUND LINMAX FOR CRT (START NEW LINE)
0218  D94F  A4 F3         STCH00 LDY CURPO       ;# OF CHAR = LINMAX ?
0219  D951  C8                   INY
0220  D952  C0 28                CPY #LINMAX     ;CURPO >= 40 CHARACTERS ?
0221  D954  90 02                BCC *+4         ;NO ,INCREMENT CURPO
0222  D956  A0 00                LDY #00         ;YES , RESET TO ZERO
0223  D958  84 F3                STY CURPO
0224  D95A  0A                   ASL A           ;TRANSFER MSB OF VIDEO FLAG TO DISPL
0225  D95B  06 F4                ASL REVER       ;GET MSB INTO CARRY
0226  D95D  6A                   ROR A           ;ACC NOW HAS VIDEO FLAG AND CHAR
0227  D95E  85 F4                STA REVER       ;RESTORE MSB OF REVER VIDEO FLAG
0228  D960  A0 00         STCH0  LDY #0
0229  D962  91 F0                STA (CURSOR),Y  ;STORE INTO DISPLAY RAM
0230  D964  E6 F0         STCH1  INC CURSOR      ;INCR CURSOR
0231  D966  D0 02                BNE STCH3
0232  D968  E6 F1         STCH2  INC CURSOR+1
0233  D96A  A5 F0         STCH3  LDA CURSOR      ;SEE IF > SCRMAX (SCREEN MAX)
```

```
LINE #  LOC    CODE        LINE

0234   D96C  C9 80          CMP #<SCRMAX
0235   D96E  D0 06          BNE STCH4
0236   D970  A5 F1          LDA CURSOR+1
0237   D972  29 0F          AND #$0F
0238   D974  C9 02          CMP #>SCRMAX
0239   D976  60          STCH4  RTS

0241   D977                ;TRANSFER CURSOR TO ACTUAL CURSOR IN 6845 & TO AIM CURSOR
0242   D977  A9 0E       TRANSF LDA #14          ;ACCESS THE ADDRESS REG
0243   D979  8D 00 80         STA RS            ;REGISTER SELECT
0244   D97C  A5 F1            LDA CURSOR+1      ;PUT CURSOR IN CURSOR REG (6845)
0245   D97E  29 0F            AND #$0F
0246   D980  8D 01 80         STA CURHIG
0247   D983  A9 0F            LDA #15
0248   D985  8D 00 80         STA RS
0249   D988  A5 F0            LDA CURSOR
0250   D98A  8D 01 80         STA CURLOW
0251   D98D  60               RTS


0253   D98E  BD 04 01    DELETE LDA $104,X       ;DONT DECREMENT BEYOND ZERO
0254   D991  C9 14            CMP #20           ;Y REG >= 20 ?
0255   D993  90 02            BCC *+4           ;NO ,RESET CURPO TO THAT VALUE
0256   D995  A9 13            LDA #19           ;YES ,RESET TO 19 TO SEE DELETES
0257   D997  8D 15 A4         STA CURPO2
0258   D99A  C6 F2            DEC CURP2         ;DECR OTHER POINTER
0259   D99C  C6 F3            DEC CURPO         ;WRAP AROUND ZERO
0260   D99E  10 04            BPL DEL2
0261   D9A0  A9 27            LDA #LINMAX-1     ;RESET TO LINMAX-1
0262   D9A2  85 F3            STA CURPO
0263   D9A4  A9 E8       DEL2   LDA #$E8         ;STORE NEW RTN ADDRESS FOR OUTDP1
0264   D9A6  9D 07 01         STA $107,X
0265   D9A9  A9 04            LDA #$04          ;RTN TO PSL00+3 WITH NEW POINTER
0266   D9AB  9D 06 01         STA $106,X
0267   D9AE  A5 F2            LDA CURP2         ;NEW POINTER TO SAVED ACC
0268   D9B0  9D 05 01         STA $105,X
0269   D9B3  20 BD D9         JSR DECREM        ;DECREMENT CURSOR
0270   D9B6  98               TYA               ;CLEAR LAST CHAR
0271   D9B7  A0 00            LDY #00
0272   D9B9  91 F0            STA (CURSOR),Y
0273   D9BB  C8               INY               ;SET Z FLAG TO 1
0274   D9BC  60               RTS


0276   D9BD  C6 F0       DECREM DEC CURSOR       ;DECREMENT CURSOR AND CURSOR+1
0277   D9BF  A5 F0            LDA CURSOR
0278   D9C1  C9 FF            CMP #$FF          ;WAS IT 00 LAST TIME ?
0279   D9C3  D0 02            BNE DECR2
0280   D9C5  C6 F1       DECR1  DEC CURSOR+1
0281   D9C7  A9 8F       DECR2  LDA #>REFRAM-1   ;CURSOR < $9000 ?
0282   D9C9  C5 F1            CMP CURSOR+1
0283   D9CB  60               RTS
0284   D9CC                   .END
```

```
TV INTERFACE SUBROUTINE.....PAGE 0007

LINE # LOC      CODE        LINE

ERRORS = 0000 <0000>
```

```
SYMBOL TABLE

SYMBOL    VALUE

CLR       D828    CLRSCR    D89D    CRLFTV    D83F    CRTV0     D844
CRTV1     D84D    CURD2     D8F3    CURDOW    D8E7    CURHIG    8001
CURLEF    D905    CURLOW    8001    CURP2     00F2    CURP0     00F3
CURP02    A415    CURRIG    D856    CURSOR    00F0    CURUP     D8F6
DECR1     D9C5    DECR2     D9C7    DECREM    D9BD    DEL2      D9A4
DELETE    D98E    DIBUFF    A430    DILINK    A406    ENTCH1    D8B9
ENTCH2    D8BB    ENTCHR    D8B6    EOS       D8A5    HOM1      D8C0
HOME      D8BD    HOMECU    D914    INIT      D804    LINMAX    0028
OUTTV     D85C    PHXY      EB9E    PLXY      EBAC    REFRAM    9000
REVER     00F4    REVIDE    D90B    RS        8000    SCR1      D8CB
SCR2      D8D5    SCRMAX    0280    SCROLL    D8C8    SPACES    D91F
STCH0     D960    STCH00    D94F    STCH1     D964    STCH2     D968
STCH3     D96A    STCH4     D976    STCHR     D936    STORE     D927
TABL      D82F    TEMP      A42D    TRANSF    D977
END OF ASSEMBLY
```


J3
AIM 65
EXPANSION
CONNECTOR
D7-D0
A15-A0
MA0-MA13
address
Data
IC1
MC6845
CRTC
IC2
CG50041-1
CHARACTER
GENERATOR
IC3
SN74LS157
MULTIPLEXER
IC4
SN74LS157
MULTIPLEXER
IC5
74LS157
MULTIPLEXER
IC6
2114
RAM
IC7
2114
RAM
IC9
8T28
IC10
8T28
IC14
7474
FF
IC16
7474
FF
IC15
7486
GATED OSC.
REVERSE VIDEO OPTION
VIDEO
ADJUST SYNC
LEVEL
ADJUST
SIGNAL
AMPLITUDE
VIDEO
OUT
R1
3.3K
R2
3.3K
R10
680
R7
1K
R8
1K
R11
220
R12
100
R9
1K
C1
270PF
C2
47PF
C3
270PF
D1
IN4148
D2
IN4148
D3
IN4148
Q1
2N2222
2K
P1
100
P2
7406
7402
7404
DISEN
HSYNC
VSYNC
CURSOR
+5V
GND

ROCKWELL INTERNATIONAL – MICROELECTRONIC DEVICES

REGIONAL SALES OFFICES

SOUTHWEST REGION, U.S.A.*

3310 Miraloma Avenue
P.O. Box 3669
Anaheim, California 92803
(714) 632-0950
TWX: 910-591-1698

NORTHWEST REGION, U. S. A.

Rockwell International
1601 Civic Center Drive, Suite 203
Santa Clara, CA 95050
408/984-6070
Telex: 171135 mission snta

Rockwell International

CENTRAL REGION, U.S.A.

Contact Robert O. Whitesell & Associates
6691 East Washington Street
Indianapolis, Indiana 46219
(317) 359-9283
Attn: Milt Gamble, Mgr. (Acting)
TWX: 810-341-3320

EASTERN REGION, U.S.A.*

Carolier Office Building
850-870 U.S. Route 1
North Brunswick, New Jersey 08902
(201) 246-3630
TWX: 710 480 6261 RECTC NBR

MIDWEST REGION, U.S.A.

1011 E. Touhy Avenue, Suite 245
Des Plaines, Illinois 60018
(312) 297-8862
Telex: 726353 Rockwell

EUROPE

Rockwell International GmbH
Microelectronic Devices
Fraunhoferstrasse 11
D-8033 Munchen-Martinsried
Germany
(089) 859-9575
Telex: 0521/2650

FAR EAST

Rockwell International Overseas Corp.
Itohpia Hirakawa-cho Bldg.
7-6 Hirakawa-cho 2-chome
Chiyoda-ku, Japan
(03) 265-8806
Telex: J22198

*Also Applications Centers

YOUR LOCAL REPRESENTATIVE